1 Kane equations - Example 1

Find the equations of the motion for the system in Fig. 1 using Kane's method. The homogeneous slender rods $OA = AB = 2L$ have the masses equals to m. The mass of the slider 3 is M_1 and the mass of the slider 4 is M_2. The linear spring R has the elastic constant k and its mass is neglected. The driver moment $\mathbf{M}_m = M_m\,\mathbf{k}$ acts on 1 at O. The initial conditions are given. The friction is neglected.

Solution

For the kinematic chain of family $f = 3$, the mobility is computed with

$$M = 3\,n - 2\,c_5 - c_4 = 3\,(4) - 2\,(5) - 0 = 2,$$

where n=4 moving links and the number of full joints is c_5=5. The system has two degrees of freedom. There are two generalized coordinate $q_1(t)$ and $q_2(t)$. The first generalized coordinte is the angle $q_1(t)$ between link 1 and the vertical axis, and the second generalized coordinate $q_2(t)$ is a linear distance from the origin O to the mass center C of slider 4.

There are two *generalized speeds* defined as

$$u_1 = \dot{q}_1 \text{ and } u_2 = \dot{q}_2.$$

Kinematics

The angular velocities of links 1 and 2 in a "fixed" cartesian reference frame (0) of unit vectors $[\imath, \jmath, \mathbf{k}]$ are

$$\boldsymbol{\omega}_{10} = \dot{q}_1\mathbf{k} = u_1\mathbf{k} \text{ and } \boldsymbol{\omega}_{20} = -\dot{q}_1\mathbf{k} = -u_1\mathbf{k}. \tag{1}$$

The angular accelerations of links 1 and 2 in (0) are

$$\boldsymbol{\alpha}_{10} = \dot{\boldsymbol{\omega}}_{10} = \ddot{q}_1\mathbf{k} = \dot{u}_1\mathbf{k}, \quad \boldsymbol{\alpha}_{20} = \dot{\boldsymbol{\omega}}_{10} = -\ddot{q}_1\mathbf{k} = -\dot{u}_1\mathbf{k}. \tag{2}$$

The position vector of the mass center C_1 of the link 1 is

$$\mathbf{r}_{C1} = L\cos q_1\imath + L\sin q_1\jmath,$$

and the position vector of the mass center C_2 of the link 2 is

$$\mathbf{r}_{C2} = 3L\cos q_1\imath + L\sin q_1\jmath.$$

The position vector of the mass center B of the link 3 is

$$\mathbf{r}_B = 4L\cos q_1 \mathbf{\imath},$$

and the position vector of the mass center C of the link 4 is

$$\mathbf{r}_C = q_2 \mathbf{\imath}.$$

The linear velocities of the mass centers are

$$\begin{aligned}
\mathbf{v}_{C1} &= \dot{\mathbf{r}}_{C1} = -Lu_1 \sin q_1 \mathbf{\imath} + Lu_1 \cos q_1 \mathbf{\jmath},\\
\mathbf{v}_{C2} &= \dot{\mathbf{r}}_{C2} = -3Lu_1 \sin q_1 \mathbf{\imath} + Lu_1 \cos q_1 \mathbf{\jmath},\\
\mathbf{v}_B &= \dot{\mathbf{r}}_B = -4Lu_1 \sin q_1 \mathbf{\imath},\\
\mathbf{v}_C &= \dot{\mathbf{r}}_C = u_2 \mathbf{\imath}.
\end{aligned} \tag{3}$$

The linear accelerations of the mass centers are

$$\begin{aligned}
\mathbf{a}_{C1} &= \dot{\mathbf{v}}_{C1} = -L(\dot{u}_1 \sin q_1 + u_1^2 \cos q_1)\mathbf{\imath} + L(\dot{u}_1 \cos q_1 - u_1^2 \sin q_1)\mathbf{\jmath},\\
\mathbf{a}_{C2} &= \dot{\mathbf{v}}_{C2} = -3L(\dot{u}_1 \sin q_1 + u_1^2 \cos q_1)\mathbf{\imath} + L(\dot{u}_1 \cos q_1 - u_1^2 \sin q_1)\mathbf{\jmath},\\
\mathbf{a}_B &= \dot{\mathbf{v}}_B = -4L(\dot{u}_1 \sin q_1 + u_1^2 \cos q_1)\mathbf{\imath},\\
\mathbf{a}_C &= \dot{\mathbf{v}}_C = \dot{u}_2 \mathbf{\imath}.
\end{aligned} \tag{4}$$

Generalized inertia forces

The generalized inertia forces for a rigid body RB are

$$F_r^* = \frac{\partial \mathbf{v}_{CG}}{\partial u_r} \cdot \mathbf{F}_{\text{in}} + \frac{\partial \boldsymbol{\omega}}{\partial u_r} \cdot \mathbf{T}_{\text{in}}, \tag{5}$$

where $\mathbf{v}_{CG}$ is the velocity of the mass center RB in (0), and $\boldsymbol{\omega} = \omega_x \mathbf{\imath} + \omega_y \mathbf{\jmath} + \omega_z \mathbf{k}$ is the angular velocity of RB in (0).
The inertia force for the rigid body RB is

$$\mathbf{F}_{\text{in}} = -M\,\mathbf{a}_{CG}, \tag{6}$$

where M is the mass of RB, and $\mathbf{a}_{CG}$ is the acceleration of the mass center of RB in (0).
The inertia torque $\mathbf{T}_{\text{in}}$ for RB is

$$\mathbf{T}_{\text{in}} = -\boldsymbol{\alpha} \cdot \bar{I} - \boldsymbol{\omega} \times (\bar{I} \cdot \boldsymbol{\omega}), \tag{7}$$

where $\boldsymbol{\alpha} = \dot{\boldsymbol{\omega}} = \alpha_x \boldsymbol{\imath} + \alpha_y \boldsymbol{\jmath} + \alpha_z \mathbf{k}$ is the angular acceleration of RB in (0), and $\bar{I} = (I_x \boldsymbol{\imath})\boldsymbol{\imath} + (I_y \boldsymbol{\jmath})\boldsymbol{\jmath} + (I_z \mathbf{k})\mathbf{k}$ is the central inertia dyadic of RB. The central principal axes of RB are parallel to $\boldsymbol{\imath}$, $\boldsymbol{\jmath}$, $\mathbf{k}$ and the associated moments of inertia have the values I_x, I_y, I_z, respectively.

- Rigid body 1:

$$\begin{aligned} \mathbf{F}_{\text{in }1} &= -m\mathbf{a}_{C1} = mL(\dot{u}_1 \sin q_1 + u_1^2 \cos q_1)\boldsymbol{\imath} - mL(\dot{u}_1 \cos q_1 - u_1^2 \sin q_1)\boldsymbol{\jmath}, \\ \mathbf{T}_{\text{in }1} &= -\boldsymbol{\alpha}_{10} \cdot \bar{I}_1 = -\boldsymbol{\alpha}_{10} I_{1z} = -\frac{mL^2}{12}\dot{u}_1 \mathbf{k}. \end{aligned} \tag{8}$$

- Rigid body 2:

$$\begin{aligned} \mathbf{F}_{\text{in }2} &= -m\mathbf{a}_{C2} = 3mL(\dot{u}_1 \sin q_1 + u_1^2 \cos q_1)\boldsymbol{\imath} - mL(\dot{u}_1 \cos q_1 - u_1^2 \sin q_1)\boldsymbol{\jmath}, \\ \mathbf{T}_{\text{in }2} &= -\boldsymbol{\alpha}_{20} \cdot \bar{I}_2 = -\boldsymbol{\alpha}_{20} I_{2z} = \frac{mL^2}{12}\dot{u}_1 \mathbf{k}. \end{aligned} \tag{9}$$

- Rigid body 3:

$$\begin{aligned} \mathbf{F}_{\text{in }3} &= -M_1 \mathbf{a}_B = 4M_1 L(\dot{u}_1 \sin q_1 + u_1^2 \cos q_1), \boldsymbol{\imath} \\ \mathbf{T}_{\text{in }3} &= \mathbf{0}. \end{aligned} \tag{10}$$

- Rigid body 4:

$$\begin{aligned} \mathbf{F}_{\text{in }4} &= -M_2 \mathbf{a}_C = -M_2 \dot{u}_2 \boldsymbol{\imath}, \\ \mathbf{T}_{\text{in }4} &= \mathbf{0}. \end{aligned} \tag{11}$$

The generalized inertia forces associated to q_1 and q_2 are

$$\begin{aligned} F_1^* &= \frac{\partial \mathbf{v}_{C1}}{\partial u_1} \cdot \mathbf{F}_{\text{in }1} + \frac{\partial \boldsymbol{\omega}_{10}}{\partial u_1} \cdot \mathbf{T}_{\text{in }1} + \\ &\quad \frac{\partial \mathbf{v}_{C2}}{\partial u_1} \cdot \mathbf{F}_{\text{in }2} + \frac{\partial \boldsymbol{\omega}_{20}}{\partial u_1} \cdot \mathbf{T}_{\text{in }2} + \\ &\quad \frac{\partial \mathbf{v}_{B}}{\partial u_1} \cdot \mathbf{F}_{\text{in }3} + \frac{\partial \mathbf{v}_{C}}{\partial u_1} \cdot \mathbf{F}_{\text{in }4}, \\ F_2^* &= \frac{\partial \mathbf{v}_{C1}}{\partial u_2} \cdot \mathbf{F}_{\text{in }1} + \frac{\partial \boldsymbol{\omega}_{10}}{\partial u_2} \cdot \mathbf{T}_{\text{in }1} + \end{aligned}$$

$$\frac{\partial \mathbf{v}_{C2}}{\partial u_2} \cdot \mathbf{F}_{\text{in }2} + \frac{\partial \boldsymbol{\omega}_{20}}{\partial u_2} \cdot \mathbf{T}_{\text{in }2} + \\ \frac{\partial \mathbf{v}_{B}}{\partial u_2} \cdot \mathbf{F}_{\text{in }3} + \frac{\partial \mathbf{v}_{C}}{\partial u_2} \cdot \mathbf{F}_{\text{in }4}, \tag{12}$$

or

$$\begin{aligned} F_1^* &= \left[-6.1666mL^2 + 8M_1L^2(\cos 2q_1 - 1) + 4mL^2\cos 2q_1\right]\dot{u}_1 \\ &\quad -4L^2(m+2M_1)u_1^2\sin 2q_1, \\ F_2^* &= -M_2\dot{u}_2. \end{aligned} \tag{13}$$

Generalized active forces

The gravitational forces are

$$\begin{aligned} &\mathbf{G}_1 = mg\boldsymbol{\imath}, \text{ acts at } C_1, \\ &\mathbf{G}_2 = mg\boldsymbol{\imath}, \text{ acts at } C_2, \\ &\mathbf{G}_3 = M_1g\boldsymbol{\imath}, \text{ acts at } B, \\ &\mathbf{G}_4 = M_2g\boldsymbol{\imath}, \text{ acts at } C, \end{aligned} \tag{14}$$

where $g = 9.81$ m/s^2 is gravitational acceleration.
The elastic forces that act on the siders 3 and 4 are given by

$$\begin{aligned} \mathbf{F}_e^{\text{link }3} &= \mathbf{F}_{eB} = -k[(\mathbf{r}_B - \mathbf{r}_C) - (\mathbf{r}_{B0} - \mathbf{r}_{C0})] = \\ &\quad k[-4L(\cos q_1 - q_2) - (4L\cos q_{10} - q_{20})]\boldsymbol{\imath}, \\ \mathbf{F}_e^{\text{link }4} &= \mathbf{F}_{eC} = -k[(\mathbf{r}_C - \mathbf{r}_B) - (\mathbf{r}_{C0} - \mathbf{r}_{B0}) = -\mathbf{F}_{eB}, \end{aligned} \tag{15}$$

where $q_{10} = q_1(0)$ and $q_{20} = q_2(0)$ are the initial generalized coordinates at $t = 0$.

The generalized active forces associated to q_1 and q_2 are

$$\begin{aligned} F_1 &= \frac{\partial \boldsymbol{\omega}_{10}}{\partial u_1} \cdot \mathbf{M}_m + \frac{\partial \mathbf{v}_{C1}}{\partial u_1} \cdot \mathbf{G}_1 + \frac{\partial \mathbf{v}_{C2}}{\partial u_1} \cdot \mathbf{G}_2 + \\ &\quad \frac{\partial \mathbf{v}_{B}}{\partial u_1} \cdot \mathbf{G}_3 + \frac{\partial \mathbf{v}_{B}}{\partial u_1} \cdot \mathbf{F}_{eB} + \frac{\partial \mathbf{v}_{C}}{\partial u_1} \cdot \mathbf{G}_4 + \frac{\partial \mathbf{v}_{C}}{\partial u_1} \cdot \mathbf{F}_{eC}, \end{aligned}$$

$$
\begin{aligned}
F_2 &= \frac{\partial \boldsymbol{\omega}_{10}}{\partial u_2} \cdot \mathbf{M}_m + \frac{\partial \mathbf{v}_{C1}}{\partial u_2} \cdot \mathbf{G}_1 + \frac{\partial \mathbf{v}_{C2}}{\partial u_2} \cdot \mathbf{G}_2 + \\
&\quad \frac{\partial \mathbf{v}_B}{\partial u_2} \cdot \mathbf{G}_3 + \frac{\partial \mathbf{v}_B}{\partial u_2} \cdot \mathbf{F}_{eB} + \frac{\partial \mathbf{v}_C}{\partial u_2} \cdot \mathbf{G}_4 + \frac{\partial \mathbf{v}_C}{\partial u_2} \cdot \mathbf{F}_{eC}, \qquad (16)
\end{aligned}
$$

or

$$
\begin{aligned}
F_1 &= M_m - 4gL(m + M_1)\sin q_1 + 4kL(4L\cos q_1 - 4L\cos q_{10} - q_2 + q_{20})\sin q_1, \\
F_2 &= -M_2 g + 4kL(\cos q_1 - \cos q_{10}) - k(q_2 - q_{20}). \qquad (17)
\end{aligned}
$$

The Kane's dynamical equations are

$$
F_r^* + F_r = 0, \quad r = 1, 2. \qquad (18)
$$

The solution of the system is obtained from Kane's dynamical relations Eq. (18) and from kinematical relations Eq. (1) with the initial conditions $q_{10} = q_1(0)$, $q_{20} = q_2(0)$, $u_{10} = u_1(0)$, and $u_{20} = u_2(0)$.

```
Off[General::spell1];
Off[General::spell];

(*position analysis*)
rC1={  L Cos[q1[t]], L Sin[q1[t]], 0};
rC2={3 L Cos[q1[t]], L Sin[q1[t]], 0};
rB={4 L Cos[q1[t]], 0, 0};
rC={q2[t],0,0};

(*velocity analysis*)
vC1=D[rC1,t];
vC2=D[rC2,t];
vB=D[rB,t];
vC=D[rC,t];

(*angular velocity analysis*)
w10={0,0,q1'[t]};
w20={0,0,-q1'[t]};

(*acceleration analysis*)
aC1=D[vC1,t];
aC2=D[vC2,t];
aB=D[vB,t];
aC=D[vC,t];

(*angular acceleration analysis*)
alpha10=D[w10,t];
alpha20=D[w20,t];

(*generalized inertia force*)

I1z=m*L^2/12.;
I2z=m*L^2/12.;

Fin1=D[vC1,q1'[t]].(- m aC1)+D[w10,q1'[t]].(- I1z alpha10)+
     D[vC2,q1'[t]].(- m aC2)+D[w20,q1'[t]].(- I2z alpha20)+
     D[vB,q1'[t] ].(- M1 aB)+D[vC,q1'[t]].(- M2 aC);

Fin2=D[vC1,q2'[t]].(- m aC1)+D[w10,q2'[t]].(- I1z alpha10)+
     D[vC2,q2'[t]].(- m aC2)+D[w20,q2'[t]].(- I2z alpha20)+
     D[vB,q2'[t] ].(- M1 aB)+D[vC,q2'[t]].(- M2 aC);

(*generalized active force*)

(*gravitational forces*)
G1={m g,0,0};
G2={m g,0,0};
G3={M1 g,0,0};
G4={M2 g,0,0};

(*elastic forces*)
rB0={4 L Cos[q1[0]], 0, 0};
rC0={q2[0],0,0};
```

```
FeB=-k ((rB-rC)-(rB0-rC0));
FeC=-k ((rC-rB)-(rC0-rB0));

(*driver torque*)
Mm={0,0,Mmot};

F1=D[w10,q1'[t]].Mm+D[vC1,q1'[t]].G1+D[vC2,q1'[t]].G2+
   D[vB,q1'[t]].G3+ D[vB,q1'[t]].FeB+
   D[vC,q1'[t]].G4+ D[vC,q1'[t]].FeC;

F2=D[w10,q2'[t]].Mm+D[vC1,q2'[t]].G1+D[vC2,q2'[t]].G2+
   D[vB,q2'[t]].G3+ D[vB,q2'[t]].FeB+
   D[vC,q2'[t]].G4+ D[vC,q2'[t]].FeC;

(*Kane's equations*)

eq1=Fin1+F1;
eq2=Fin2+F2;
```


0
O
(0)
J
$\mathbf{M}_m$
q_1
C_1
1
$m\mathbf{g}$
A
C_2
2
$M_1\mathbf{g}$
q_2
$m\mathbf{g}$
3
B
$\mathbf{F}_{eB}$
R
$\mathbf{F}_{eC}$
0
C
4
$M_2\mathbf{g}$
ı

Figure 1